FIJI (ImageJ) Macro to Automate Counting of *Perkinsus spp.*

Teresa Vaillancourt, Ph.D, SERC volunteer, December 2021

Accurate quantification of *Perkinsus* infection of oysters may be obtained by the whole body burden assay. That assay is possible because of the remarkable ability of stainable Perkinsus cell walls to survive prolonged immersion in a 2M NaOH solution at 60° C, a solution that otherwise dissolves virtually all other cell and tissue components of the parasite and its host. Following processing, *Perkinsus* samples can be accurately counted using FIJI, an enhanced version of the ImageJ image processing program developed by NIH.

The following macro is based on automatic separation of putative Perkinsus hypnospores and background by using the Phansalker local threshold algorithm. This macro may be used to analyze single images or an entire directory of images for whole body burden assays and for assays of filtered habitat water and/or feces.

What follows is an annotated version written in the ImageJ macro language; the actual commands are in **RED**. Adjacent images represent the working image resulting from the described segment of code.

```
/*
This macro is optimized for counting samples of RFTM
incubated, NaOH digested, Lugol's stained samples that have
been imaged with high resolution color micrographs.
Dimensional parameters are in "pixel" units.

The first steps record the name of the open image file (adjacent
example), and create copies for later use.
*/
FileName = File.nameWithoutExtension;
rename(FileName);
run("Duplicate...", "title=Copy1");
run("Duplicate...", "title=Copy2");
```



```
/*
The image is then split into RGB luminance channels. The
Green and Blue channels are discarded, and the Red
luminance channel (see adjacent image) becomes the working
image – it provides the best contrast separation between the
reddish background of Lugol's stained samples and the
blue-black Perkinsus hypnospores.
*/
selectWindow(FileName);
run("Split Channels");
```



```
close();
close();
/*
```
The working image is segmented into “background” and “target” pixels by using the Phansalkar local threshold algorithm. The result is a black and white image. Perkinsus hypnospores (as well as similarly sized debris) are reliably segregated when the algorithm’s radius variable is set to 100 pixels.



The watershed algorithm is used to automatically detect and draw a line between instances of clustered Perkinsus, although it is not 100% accurate – see later note regarding manual adjustment of final count.

```
*/
rename(FileName);
run("Auto Local Threshold", "method=Phansalkar radius=100 parameter_1=0 parameter_2=0
white");
run("Gaussian Blur...", "sigma=3");
run("Convert to Mask");
run("Watershed");

/*
```
Perkinsus hypnospores are circular or oval in cross-section, distinguishing them from irregularly shaped debris. Particles having diameters (in pixel units) in the expected range for Perkinsus hypnospores and having circular cross-sections are identified, enumerated, measured, tabulated, and masked using the “Analyze Particles” algorithm (adjacent image). Particles on image margins are excluded.



```
*/
run("Set Measurements...", "display redirect=None decimal=0");
run("Analyze Particles...", "size=50-1000000 circularity=0.85-1.00 show=[Overlay Masks]
summarize include");
run("Overlay Options...", "stroke=cyan width=0 fill=cyan set apply");
run("Magenta");
run("Flatten");
run("Copy");
close();
close();


/*
```
A montage of the original and “analyzed” image is created to simplify the process of authentication and correction of the automated count. In the “analyzed” portion of the montage,

particles masked in **CYAN** were counted, those masked in **MAGENTA** were debris or possible Perkinsus hypnospores that did not meet the specified size and circularity criteria or that were part of clusters not separable by the “Watershed” algorithm. Upon inspection, the user may choose to modify the count by adding specific Magenta-masked particles, e.g. the two associated with debris in the upper left quadrant, or deleting miscounted Blue-masked ones.
*/

```
selectWindow("Copy2");
run("Select All");
setBackgroundColor(0,0,0);
run("Clear", "slice");
run("Paste");
selectWindow("Copy1");
run("Images to Stack", "name=Stack title=[] use");
run("Make Montage...", "columns=2 rows=1 scale=1");
close ("Stack");
rename(FileName);
Table.deleteColumn("%Area");
Table.deleteColumn("Total Area");
```



The resulting table may be saved and opened in a spreadsheet program for further analysis.



Slice	Count
IMG_0002	39
IMG_0003	39
IMG_0004	90
IMG_0005	83
IMG_0006	98
IMG_0007	72
IMG_0008	76
IMG_0009	59

Implementation

To save the macro, open a new text window in FIJI (**File>New>Text Window**), copy and paste the following text, and save the file with an ijm extension, e.g., Count.ijm.

```
/*
Count.ijm

This macro is optimized for counting samples of RFTM incubated, NaOH digested, Lugol's stained samples that have been imaged with high resolution color micrographs. Dimensional parameters are in “pixel” units

Marine Disease Ecology Laboratory, Smithsonian Environmental Research Center
December 2021
License: BSD3

 Copyright 2022

 Redistribution and use in source and binary forms, with or without modification, are permitted provided that the following conditions are met:
 1. Redistributions of source code must retain the above copyright notice, this list of conditions and the following disclaimer.
 2. Redistributions in binary form must reproduce the above copyright notice, this list of conditions and the following disclaimer in the documentation and/or other materials provided with the distribution.
 3. Neither the name of the copyright holder nor the names of its contributors may be used to endorse or promote products derived from this software without specific prior written permission.
 THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.

*/
FileName = File.nameWithoutExtension;
rename(FileName);
run("Duplicate...", "title=Copy1");
run("Duplicate...", "title=Copy2");
selectWindow(FileName);
run("Split Channels");
close();
close();
rename(FileName);
run("Auto Local Threshold", "method=Phansalkar radius=100 parameter_1=0 parameter_2=0 white");
run("Gaussian Blur...", "sigma=3");
run("Convert to Mask");
run("Watershed");
run("Set Measurements...", "display redirect=None decimal=0");
run("Analyze Particles...", "size=50-1000000 circularity=0.85-1.00 show=[Overlay Masks] summarize include");
run("Overlay Options...", "stroke=cyan width=0 fill=cyan set apply");
run("Magenta");
run("Flatten");
run("Copy");
close();
close();
selectWindow("Copy2");
run("Select All");
setBackgroundColor(0,0,0);
run("Clear", "slice");
run("Paste");
selectWindow("Copy1");
run("Images to Stack", "name=Stack title=[] use");
run("Make Montage...", "columns=2 rows=1 scale=1");
close ("Stack");
rename(FileName);
Table.deleteColumn("%Area");
Table.deleteColumn("Total Area");
Table.deleteColumn("Average Size");
```

To analyze a single image, first close any previously opened images (**File>Close All**), then open the image of interest and select **“Run”** in the text editor window containing the macro code.

To run the macro in batch mode, open the batch processor (**Process>Batch>Macro**), replace any existing content in the code window with the macro code above, select the directory containing image files to be analyzed and the directory to receive the analyzed montage files, then press **Process**. Note that the **Summary** tables should be renamed to reflect the experiment’s designation and saved manually, if so desired.